Appendices - Computer programs and output used in the Examples

Appendix 1a

SAS Program

```
data one;
  do resp='comp','subs','some','none';
    resp2=1*(resp='comp')+2*(resp='subs')
          +3*(resp='some')+4*(resp='none');
  do drug='a','b';
    druga=(drug='a');
    input count @@; output;
  end; end; cards;
5 2 10 5 7 11 3 7
;
proc freq order=data;
  weight count;
  tables drug*resp/chisq cmh nopercent nocol;
  exact fisher chisq;
run;
```

SAS Output

The FREQ Procedure

drug resp

Frequency Row Pct	comp	subs	some	none	Total
a	5 20.00	10 40.00	7 28.00	3 12.00	25
b	2 8.00	5 20.00	11 44.00	7 28.00	25
Total	7	15	18	10	50

Statistics for Table of drug by resp

Statistic	DF	Value	Prob
Chi-Square	3	5.4413	0.1422
Likelihood Ratio Chi-Square	3	5.5693	0.1346
Mantel-Haenszel Chi-Square	1	4.8165	0.0282
Phi Coefficient		0.3299	
Contingency Coefficient		0.3133	
Cramer's V		0.3299	

WARNING: 25% of the cells have expected counts less than 5.
(Asymptotic) Chi-Square may not be a valid test.

Pearson Chi-Square Test

Chi-Square	5.4413
DF	3
Asymptotic Pr > ChiSq	0.1422
Exact Pr >= ChiSq	0.1546

Likelihood Ratio Chi-Square Test

Chi-Square	5.5693
DF	3
Asymptotic Pr > ChiSq	0.1346
Exact Pr >= ChiSq	0.1700

Mantel-Haenszel Chi-Square Test

Chi-Square	4.8165
DF	1
Asymptotic Pr > ChiSq	0.0282
Exact Pr >= ChiSq	0.0389

Fisher's Exact Test

Table Probability (P)	0.0019
Pr <= P	0.1618

Sample Size = 50

Summary Statistics for drug by resp

Cochran-Mantel-Haenszel Statistics (Based on Table Scores)

Statistic	Alternative Hypothesis	DF	Value	Prob
1	Nonzero Correlation	1	4.8165	0.0282
2	Row Mean Scores Differ	1	4.8165	0.0282
3	General Association	3	5.3324	0.1490

Total Sample Size = 50

Appendix 1b

SAS Program

```
proc logistic order=data;
  weight count;
  model resp2=druga;
run;
```

SAS Output

The LOGISTIC Procedure

Score Test for the Proportional Odds Assumption

Chi-Square	DF	Pr > ChiSq
0.2683	2	0.8745

Testing Global Null Hypothesis: BETA=0

Test	Chi-Square	DF	Pr > ChiSq
Likelihood Ratio	5.3107	1	0.0212
Score	5.1209	1	0.0236
Wald	5.0910	1	0.0241

Analysis of Maximum Likelihood Estimates

Parameter	DF	Estimate	Standard Error	Wald Chi-Square	Pr > ChiSq
Intercept 1	1	-2.5625	0.5445	22.1448	<.0001
Intercept 2	1	-0.8718	0.4114	4.4914	0.0341
Intercept 3	1	0.8957	0.4135	4.6924	0.0303
druga	1	1.2205	0.5409	5.0910	0.0241

Odds Ratio Estimates

Effect	Point Estimate	95% Wald Confidence Limits	
druga	3.389	1.174	9.784

Appendix 2a

SAS Program

```
data one;
  do age=0.5,1.5,2.5,3.5,4.5,7.5,12.5,17.5,22.5,30,40,50,60;
    input yes n @@; fraction=yes/n; output;
  end; datalines;
1 13 3 7 2 12 2 6 5 10 26 64 23 39
17 32 9 12 59 89 59 87 48 67 13 22
;
proc nlmixed data=one;
  parms th2=10 th3=0.5 gamma=1;
  if abs(gamma) le 0.0001 then do;
    ex=(age/th2)**th3;
  end;
  else do;
    zx=(age**gamma - 1)/gamma; zth2=(th2**gamma - 1)/gamma;
    ex=exp(th3*(zx-zth2));
  end;
  p=ex/(1+ex);
  model yes~binomial(n,p);
run;
```

SAS Output

Parameter Estimates

Parameter	Estimate	Std Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
th2	9.7222	2.5099	13	3.87	0.0019	0.05	4.3000	15.1445
th3	0.8840	0.4492	13	1.97	0.0708	0.05	-0.08640	1.8544
gamma	-0.1795	0.2236	13	-0.80	0.4365	0.05	-0.6625	0.3035

Appendix 2b

SAS Program

```
proc nlmixed data=one;
  parms th2=10 th3=1;
  t=(age/th2)**th3;
  p=t/(1+t);
  model yes~binomial(n,p);
run;
```

SAS Output

Parameter Estimates

Parameter	Estimate	Std Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
th2	10.9717	2.1488	13	5.11	0.0002	0.05	6.3295	15.6139
th3	0.5870	0.1017	13	5.77	<.0001	0.05	0.3673	0.8067

Appendix 3

SAS Program

```
data one;
  do male='correct  ','incorrect';
  do female='correct  ','incorrect';
    input count @@; output;
  end; end; datalines;
6 10 19 1
;
proc freq order=data;
  tables male*female/nocol norow nopercent agree;
  weight count;
run;
```

SAS Output

The FREQ Procedure

Table of male by female

male Frequency	female correct	incorrect	Total
correct	6	10	16
incorrect	19	1	20
Total	25	11	36

Statistics for Table of male by female

McNemar's Test

Statistic (S)	2.7931
DF	1
Pr > S	0.0947

Simple Kappa Coefficient

Kappa	-0.5444
ASE	0.1413
95% Lower Conf Limit	-0.8214
95% Upper Conf Limit	-0.2673

Sample Size = 36

Appendix 4a

Minitab Output

One-way ANOVA: y versus trt

```
Source  DF     SS    MS     F      P
trt      2  132.8  66.4  3.10  0.061
Error   27  577.5  21.4
Total   29  710.3

S = 4.625   R-Sq = 18.69%   R-Sq(adj) = 12.67%

                          Individual 95% CIs For Mean Based on
                          Pooled StDev
Level   N    Mean  StDev  ----+---------+---------+---------+-----
1      10  75.830  2.448                  (---------*---------)
2      10  72.940  7.313         (---------*---------)
3      10  70.690  2.166  (---------*---------)
                          ----+---------+---------+---------+-----
                           69.0      72.0      75.0      78.0
Pooled StDev = 4.625
```

Appendix 4b

SAS Program

```
proc iml;
  start n2lla(th) global(y1,y2,y3,n1,n2,n3);
    mu1=th[1]; mu2=th[2]; mu3=th[3]; sig1=th[4]; sig2=th[5]; sig3=th[6];
    var1=sig1*sig1; var2=sig2*sig2; var3=sig3*sig3;
    y1d=y1-mu1; y2d=y2-mu2; y3d=y3-mu3;
    expr=n1*log(var1)+n2*log(var2)+n3*log(var3)
        +(1/var1)*t(y1d)*y1d+(1/var2)*t(y2d)*y2d+(1/var3)*t(y3d)*y3d;
    return(expr);
  finish n2lla;
  th0a={70 70 70  5  5  5}; cona={40 40 40  1  1  1, 90 90 90  9  9  9};
  start n2llb(th) global(y1,y2,y3,n1,n2,n3);
    mu=th[1]; sig1=th[2]; sig2=th[3]; sig3=th[4];
    var1=sig1*sig1; var2=sig2*sig2; var3=sig3*sig3;
    y1d=y1-mu; y2d=y2-mu; y3d=y3-mu;
    expr=n1*log(var1)+n2*log(var2)+n3*log(var3)
        +(1/var1)*t(y1d)*y1d+(1/var2)*t(y2d)*y2d+(1/var3)*t(y3d)*y3d;
    return(expr);
  finish n2llb;
  th0b={70  5  5  5}; conb={40  1  1  1, 90 12  12 12};
  y1={79.6,71.3,77.7,75.2,77.7,74.5,75.0,77.8,76.0,73.5};
  y2={76.1,76.2,74.1,76.1,62.2,67.9,63.3,72.3,87.6,73.6};
  y3={68.4,70.6,70.8,69.7,71.8,73.4,72.7,73.0,66.5,70.0};
  n1=nrow(y1); n2=nrow(y2); n3=nrow(y3); opt={.,0};
  call nlptr(rc,thetaa,"n2lla",th0a,opt,cona); minnlla=n2lla(thetaa);
  call nlptr(rc,thetab,"n2llb",th0b,opt,conb); minnllb=n2llb(thetab);
  teststat=minnllb-minnlla; pval=1-probchi(teststat,2); print teststat pval;
quit;
```

SAS Output

TESTSTAT	PVAL
16.715658	0.0002346

Appendix 5a

SAS Program

```
****  Logistic method  ****;

data one;
  do drugb=0,1;
  do week=2,4,6,8;
    input survive @@; n=20; output;
  end; end; datalines;
16 10 6 2 18 12 10 6
;
proc nlmixed;
  parms th2a=4 th2b=6 th3a=2 th3b=2;
  th2=th2a*(drugb=0)+th2b*(drugb=1);
  th3=th3a*(drugb=0)+th3b*(drugb=1);
  t=( week /th2)**th3; den=1+t; pi=1/den;
  model survive~binomial(n,pi);
run;
proc nlmixed;
  parms th2a=4 th2b=6 th3=2;
  th2=th2a*(drugb=0)+th2b*(drugb=1);
  t=( week /th2)**th3; den=1+t; pi=1/den;
  model survive~binomial(n,pi);
run;
proc nlmixed;
  parms th2a=4 rho=1 th3=2;
  th2b=rho*th2a;
  th2=th2a*(drugb=0)+th2b*(drugb=1);
  t=( week /th2)**th3; den=1+t; pi=1/den;
  model survive~binomial(n,pi);
run;
proc nlmixed;
  parms th2a=4 th3=2; rho=1;
  th2b=rho*th2a;
  th2=th2a*(drugb=0)+th2b*(drugb=1);
  t=( week /th2)**th3; den=1+t; pi=1/den;
  model survive~binomial(n,pi);
run;
```

SAS Output

First NLMIXED (-2LL = 26.3)

Parameter Estimates

Parameter	Estimate	Std Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
th2a	3.8228	0.4452	8	8.59	<.0001	0.05	2.7961	4.8496
th2b	5.4413	0.6838	8	7.96	<.0001	0.05	3.8644	7.0182
th3a	2.3672	0.5653	8	4.19	0.0030	0.05	1.0636	3.6709
th3b	2.0340	0.5720	8	3.56	0.0074	0.05	0.7150	3.3529

Second NLMIXED (-2LL = 26.4)

Parameter Estimates

Parameter	Estimate	Std Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
th2a	3.7866	0.4570	8	8.29	<.0001	0.05	2.7326	4.8405
th2b	5.4085	0.6275	8	8.62	<.0001	0.05	3.9615	6.8555
th3	2.2087	0.4028	8	5.48	0.0006	0.05	1.2798	3.1377

Third NLMIXED (-2LL = 26.4)

Parameter Estimates

Parameter	Estimate	Std Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
th2a	3.7865	0.4570	8	8.29	<.0001	0.05	2.7326	4.8405
rho	1.4284	0.2417	8	5.91	0.0004	0.05	0.8710	1.9857
th3	2.2087	0.4028	8	5.48	0.0006	0.05	1.2798	3.1376

Fourth NLMIXED (-2LL = 31.1)

Parameter Estimates

Parameter	Estimate	Std Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
th2a	4.5283	0.3814	8	11.87	<.0001	0.05	3.6489	5.4077
th3	2.1380	0.3926	8	5.44	0.0006	0.05	1.2325	3.0434

Appendix 5b

SAS Program

```
****  Interval censored method  ****;

data one;
  do trt=0,1;
  do pt=1 to 20;
    time=0; result=0;
    if trt=0 and pt le 4 then do;  time=2; result=2; output; end;
    else if trt=0 and pt le 10 then do; time=4; result=2; output; end;
    else if trt=0 and pt le 14 then do; time=6; result=2; output; end;
    else if trt=0 and pt le 18 then do; time=8; result=2; output; end;
    else if trt=0 and pt gt 18 then do; time=8; result=1; output; end;
    else if trt=1 and pt le 2 then do;  time=2; result=2; output; end;
    else if trt=1 and pt le 8 then do;  time=4; result=2; output; end;
    else if trt=1 and pt le 10 then do; time=6; result=2; output; end;
    else if trt=1 and pt le 14 then do; time=8; result=2; output; end;
    else if trt=1 and pt gt 14 then do; time=8; result=1; output; end;
  end; end;

data two;
  set one;
  jj=1+(time gt 2)+(time gt 4)+(time gt 6);
  do per=1 to jj;
    per1=(per=1); per2=(per=2); per3=(per=3); per4=(per=4); output;
  end;

data three;
  set two;
  rr=0;
  if result=2 and ((per=1 and time le 2)
    or (per=2 and (time gt 2 and time le 4))
    or (per=3 and (time gt 4 and time le 6))
    or (per=4 and time gt 6)) then rr=1;

proc nlmixed data=three;
  parms th2a=4 th2b=5.5 th3a=2 th3b=2;
  ka1=(2/th2a)**th3a; ka2=(4/th2a)**th3a;
  ka3=(6/th2a)**th3a; ka4=(8/th2a)**th3a;
  kb1=(2/th2b)**th3b; kb2=(4/th2b)**th3b;
  kb3=(6/th2b)**th3b; kb4=(8/th2b)**th3b;
       if trt=0 and per=1 then pp=ka1/(1+ka1);
  else if trt=0 and per=2 then pp=(ka2-ka1)/(1+ka2);
  else if trt=0 and per=3 then pp=(ka3-ka2)/(1+ka3);
  else if trt=0 and per=4 then pp=(ka4-ka3)/(1+ka4);
  else if trt=1 and per=1 then pp=kb1/(1+kb1);
  else if trt=1 and per=2 then pp=(kb2-kb1)/(1+kb2);
  else if trt=1 and per=3 then pp=(kb3-kb2)/(1+kb3);
  else if trt=1 and per=4 then pp=(kb4-kb3)/(1+kb4);
  model rr~binary(pp);
run;
```

```
proc nlmixed data=three;
  parms th2a=4 th2b=5.5 th3=2;
  ka1=(2/th2a)**th3; ka2=(4/th2a)**th3;
  ka3=(6/th2a)**th3; ka4=(8/th2a)**th3;
  kb1=(2/th2b)**th3; kb2=(4/th2b)**th3;
  kb3=(6/th2b)**th3; kb4=(8/th2b)**th3;
       if trt=0 and per=1 then pp=ka1/(1+ka1);
  else if trt=0 and per=2 then pp=(ka2-ka1)/(1+ka2);
  else if trt=0 and per=3 then pp=(ka3-ka2)/(1+ka3);
  else if trt=0 and per=4 then pp=(ka4-ka3)/(1+ka4);
  else if trt=1 and per=1 then pp=kb1/(1+kb1);
  else if trt=1 and per=2 then pp=(kb2-kb1)/(1+kb2);
  else if trt=1 and per=3 then pp=(kb3-kb2)/(1+kb3);
  else if trt=1 and per=4 then pp=(kb4-kb3)/(1+kb4);
  model rr~binary(pp);
run;

proc nlmixed data=three;
  parms th2=4 th3=2;
  k1=(2/th2)**th3; k2=(4/th2)**th3; k3=(6/th2)**th3; k4=(8/th2)**th3;
       if per=1 then pp=k1/(1+k1);
  else if per=2 then pp=(k2-k1)/(1+k2);
  else if per=3 then pp=(k3-k2)/(1+k3);
  else if per=4 then pp=(k4-k3)/(1+k4);
  model rr~binary(pp);
run;
```

SAS Output

First NLMIXED (-2LL = 126.5)

Parameter Estimates

Parameter	Estimate	Std Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
th2a	3.8526	0.6488	112	5.94	<.0001	0.05	2.5672	5.1380
th2b	5.3827	1.0444	112	5.15	<.0001	0.05	3.3134	7.4521
th3a	2.3914	0.5379	112	4.45	<.0001	0.05	1.3256	3.4573
th3b	2.0689	0.5117	112	4.04	<.0001	0.05	1.0550	3.0828

Second NLMIXED (-2LL = 126.7)

Parameter Estimates

Parameter	Estimate	Std Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
th2a	3.8359	0.6794	112	5.65	<.0001	0.05	2.4898	5.1820
th2b	5.3633	0.9770	112	5.49	<.0001	0.05	3.4276	7.2990
th3	2.2325	0.3719	112	6.00	<.0001	0.05	1.4957	2.9693

Third NLMIXED (-2LL = 128.4)

Parameter Estimates

Parameter	Estimate	Std Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
th2	4.5083	0.5818	112	7.75	<.0001	0.05	3.3555	5.6610
th3	2.1830	0.3628	112	6.02	<.0001	0.05	1.4642	2.9018